

3 Movimentos com vínculos

3.3 Movimento ao longo de uma cicloide

No último capítulo vimos que o período de um pêndulo varia com a amplitude.

Há, entretanto, um sistema especial em que o período de um pêndulo é independente da amplitude. É o pêndulo cicloidal.




Fig. 3.3-1

Uma cicloide é um tipo de curva geométrica descrita por um ponto P de uma circunferência, quando o círculo avança girando, sem deslizar, sobre uma reta, p. ex. o eixo x , fixa do seu plano.

É uma curva não-algébrica e foi estudada pela primeira vez em 1599, pelo físico e astrônomo italiano Galiléu Galilei, que também inventou seu nome.

(Se o círculo rola ao longo de outro círculo, chama-se a curva descrita por um ponto, rigidamente ligado ao círculo que rola, de epicycloide o hypocycloide.)

O eixo do círculo -ou disco girante- se move com uma velocidade dada por $\mathbf{v} = -\omega \cdot \mathbf{a}$, onde \mathbf{a} é o raio do disco que roda e ω é a sua velocidade de rotação. O sinal de menos na equação decorre do fato de que a velocidade angular no sentido horário é negativa.

O ponto \mathbf{B} de cima do disco tem velocidade $\mathbf{v} - \omega \cdot \mathbf{a} = 2\mathbf{v}$, enquanto o ponto de baixo em contato com o plano tem velocidade 0. Por tanto, o ponto de contato do disco com a pista de rolamento está sempre em repouso, o que equivale a dizer que não há deslizamento. (A velocidade do ponto P é, em cada momento, uma superposição de uma rotação e de um movimento horizontal. No ponto de baixo, a velocidade de rotação e a do movimento horizontal se cancelam, dando a velocidade zero.) Geometricamente, demonstra-se que tangentes passam sempre pelo ponto \mathbf{B} , confira com parágrafo 3.3.3.

(O pedal na borda da roda dentada de uma bicicleta descreve uma cicloide.)

Preparemos agora uma animação com MuPAD. Precisamos, para fazer isso, de uma descrição paramétrica da curva.

Da figura 3.3-1 percebemos que o arco PM é igual a $OM = a \cdot \theta$. O ângulo PCM = θ (théta) é chamado de ângulo do rolamento. Vemos que

$$x = ON = OM - NM = a(\theta - \text{sen } \theta)$$

$$y = NP = MC + AC = a(1 - \text{cos } \theta) \quad (3.3-1)$$

Para praticar o uso do MuPAD, fazemos primeiramente uma **cicloide** sem animação, em seguida vamos ver uma animação. O raio é em ambos os casos igual a um.

```
plot(plot::Curve2d([t-sin(t), 1-cos(t)],
t = 0..2*PI),
plot::Point2d([t-sin(t),1-cos(t)], t = 0..2*PI,
PointSize = 2*unit::mm,Color=RGB::Red),
plot::Circle2d(1, [t,1],t = 0..2*PI),
Scaling=Constrained)
```




Fig. 3.3-2

Animação:

Para a animação diferenciamos entre o raio r do círculo geratriz e a distância a do ponto P do centro C do círculo móvel, pois a pode ser diferente de r . As fórmulas (3.3-1) têm, neste caso geral, a forma

$x(\theta) = r\theta - a\sin(\theta)$ e $y(\theta) = r - a\cos(\theta)$. Nos programas substituímos θ por t .

(Na internet pode-se encontrar numerosos artigos sobre a cicloide. O site

<http://www.mathcurve.com/index.htm>

é especialmente bem feito e instrutivo.

Também muito bem, e em alemão !:

<http://did.mat.uni-bayreuth.de/geonet/beispiele/bernoulli/inhalt.html#top>)

- `reset()` :


```

a:=0.8: r:=1:
/*ciclóide "abreviada" ou "reduzida" porque a < r;
a é a distância PC e r é o raio do círculo.
Toma-se r fixo e varia-se a: a=r ou a>r ou a<r*/
plot(plot::Curve2d([r*t-a*sin(t),r-a*cos(t)], t=0..tmax,
tmax = 0..3*PI),
plot::Point2d([r*t-a*sin(t),r-a*cos(t)],
t = 0..3*PI,
PointSize = 2*unit::mm,Color=RGB::Green),
plot::Circle2d(r, [t,r],Color=RGB::OliveGreenDark,
t = 0..3*PI),// no caso geral: [rwt,r]
plot::Line2d([t,r],[r*t-a*sin(t),r-a*cos(t)],
Color=RGB::Red,t = 0..3*PI),
Scaling=Constrained)

```




Fig. 3.3-3

A animação a seguir é uma adaptação do Programa 3 do capítulo 2.2 para o caso de uma massa m que desce sem atrito ao longo de uma **ciclóide invertida** e que segue executando oscilações. Deve-se imaginar que a ciclóide é feita de um arame ideal, rígido e sem atrito.

O programa mostra a **velocidade** e a **aceleração** em cada ponto da trajetória.

```

x:=t->t+sin(t) :
//a ciclóide invertida tem  $x = r(\theta + \sin\theta)$ 

y:=t->1-cos(t) :
curve:=plot::Curve2d([x(t),y(t)],t=-PI..PI) :

pos:=t->(x(t),y(t))://vetor posição

vel:=t->(x'(t),y'(t))://vetor velocidade

acel:=t->(x''(t),y''(t))://aceleração

scale:=1://determina o tamanho do vetor

x1:=pos(t1)[1]://início do vetor  $(x'(t),y'(t))$ 

y1:=pos(t1)[2] :

v1:=vel(t1)[1]://velocidade do ponto  $(x(t),y(t))$ 

v2:=vel(t1)[2] :

```

```

a1:=acel(t1)[1]://aceleração
a2:=acel(t1)[2]:
x2:=x1+v1*scale://extremo do vetor (x'(t),y'(t))
y2:=y1+v2*scale:
x3:=x1+a1*scale://extremo do vetor (x''(t),y''(t))
y3:=y1+a2*scale:
p:=plot::Point2d([x1,y1],t1=-PI..PI,
Color=RGB::Black,
PointSize=3*unit::mm):
ve:=plot::Arrow2d([x1,y1],[x2,y2],
t1=-PI..PI,Color=RGB::Red):
ac:=plot::Arrow2d([x1,y1],[x3,y3],
t1=-PI..PI,Color=RGB::Green):
plot(curve,p,ac,ve,
Header="Uma partícula desce ao longo de uma
ciclóide",
HeaderFont=["Arial",12,Bold],
HeaderAlignment=Center,
Footer="verde = aceleração,
vermelho=veleocidade",
FooterAlignment=Center,
Scaling = Constrained)

```

Uma partícula desce ao longo de uma cicloide




Fig. 3.3-4

3.3.1 O pêndulo cicloidal

Retomemos o assunto do último programa: Se construirmos, num plano vertical, uma trajetória com a forma de uma cicloide, aberta para cima, e deixarmos uma massa m deslizar, sem atrito, ao longo dela, então a partícula vai exercer um movimento oscilatório sob a ação da gravidade cuja amplitude vai depender do ponto em que a partícula for abandonada, mas, e isso é surpreendente, o período será sempre igual a $T = 4\pi (r/g)^{1/2}$, r = raio do círculo que gera a cicloide, e vemos que T é independente da amplitude. Foi Christian Huygens ("*Horologium oscillatorium*", 1673) quem encontrou um meio prático para fabricar um pêndulo cicloidal.

Confira as figuras 3.3-5 e 3.3-6 para mais detalhes.

Sobre a massa m que começa a deslizar, a partir do ponto A , com velocidade 0, ao longo de uma cicloide, atuam duas forças: a força de vínculo, \mathbf{N} , e a força da gravidade, $m\mathbf{g}$. \mathbf{N} é sempre perpendicular ao vínculo, o seja à trajetória cicloidal, o peso $m\mathbf{g}$ está sempre mostrando na direção- \mathbf{j} . Os vetores \mathbf{n} e \mathbf{t} são vetores unitários em direção normal e tangencial, respectivamente.




Fig. 3.3-5

A segunda lei de Newton nos dá duas equações escalares:

$$m \frac{dv}{dt} = - m g \sin(\tau) \quad (3.3-2)$$

$$m \frac{v^2}{\rho} = N - m g \cos(\tau) \quad (3.3-3)$$

$\frac{dv}{dt}$ é o módulo da aceleração tangencial, ρ é o raio de curvatura e $\frac{v^2}{\rho}$ é o módulo da aceleração normal. Mais adiante volveremos a falar sobre estas equações.

A figura 3.3-6 mostra a idéia de Huygens. Uma massa, na extremidade inferior de um fio, é obrigado, por meio de dois obstáculos de forma de cicloides, de oscilar ao longo de uma trajetória cicloidal (que a trajetória é verdadeiramente uma cicloide, pode-se demonstrar geometricamente) com um período independente da amplitude.

Para amplitudes pequenas, os obstáculos não têm muita influência sobre as oscilações e o período será praticamente igual ao período do pêndulo simples: $T = 2\pi (l/g)^{1/2}$

O mesmo resultado segue da equação $T = 4\pi (r/g)^{1/2}$, pois, no caso do pêndulo cicloidal, temos $l = 4r$. Compare com a seguinte figura 3.3-6, que foi traçada com o programa a seguir:

- `reset() //pêndulo cicloidal`
- ```

plot(plot::Curve2d([t-sin(t)+PI,-(1-cos(t))],
 Color=RGB::Black,t= -2*PI..0),
 plot::Point2d([t-sin(t)+PI,-(1-cos(t))], t = -PI-0.2..0,
 PointSize = 3*unit::mm,Color=RGB::Red),
 plot::Line2d([t-sin(t)+PI,-(1-cos(t))],[0,2],
 Color=RGB::BLUE,t = -PI-0.2..0),
 plot::Curve2d([t-sin(t),-(1-cos(t)-2)],Color=RGB::Green,
 t = -PI..PI),
 Scaling=Constrained)

```




Fig. 3.3-6

### Pêndulo cicloidal



### 3.3.2 Exercícios

Mecanicamente, uma cicloide *normal*, com  $a = r$ , é traçada por um ponto P (p.ex. uma esferográfica) que fica rigidamente conectado com um disco de raio  $r$  que rola, sem deslizar, sobre uma pista fixa. Se a distância  $a$  do ponto P do centro do disco for maior do que  $r$ ,  $a > r$ , a trajetória traçada será uma cicloide alongada. Se  $a < r$ , a cicloide é abreviada. As equações paramétricas serão nestes casos

$$x = r \cdot t - a \cdot \text{sen } t \quad \text{e} \quad y = r - a \cdot \text{cos } t$$

Faça gráficos para o três casos de cicloides.

#### Epicycloide

resulta, quando o disco girante roda sobre um círculo fixo de raio  $R$ .

As coordenadas do ponto P são, agora,

$$x = (R+r) \cos t - a \cos [t(R+r)/r] \quad \text{e} \quad y = (R+r) \text{sen } t - a \text{sen } [t(R+r)/r].$$

#### Cardióide

É uma epicycloide com  $r = R = a$ .

O comprimento dessa curva é  $L = 8r$ , a área é  $A = 6r^2\pi$ , ou seja, seis vezes o área do disco fixo.

Uma **Hipocicloide** resulta, quando o disco girante rola no interior do círculo fixo. Os parâmetros  $a$  e  $r$  mudam de signo:

$$x = (R-r) \cos t + a \cos [t(R-r)/r] \quad \text{e} \quad y = (R-r) \text{sen } t - a \text{sen } [t(R-r)/r]$$

Se fizermos nestas equações  $4r = R$ , resultará uma **Asteróide** com a representação paramétrica

$$x = R \cos^3 t \quad \text{e} \quad y = R \text{sen}^3 t$$

daí segue a seguinte representação cartesiana


$$x^{2/3} + y^{2/3} = R^{2/3}$$

Fazer gráficos dessas curvas importantes.

## Respostas:

### Os três ciclóides:

- `reset()`://programa válido para os três casos da ciclóide  
`a:=1.4:r:=1.0//a=r ou a<r ou a>r`  
`plot(`  
`plot::Curve2d([r*t-a*sin(t),r-a*cos(t)],t=0..tmax,`  
`tmax = 0..3*PI),`  
`plot::Point2d([r*t-a*sin(t),r-a*cos(t)], t = 0..3*PI,`  
`PointSize = 2*unit::mm,Color=RGB::Red),`  
`plot::Line2d([t,r],[r*t-a*sin(t),r-a*cos(t)],`  
`Color=RGB::Green,t = 0..3*PI),`  
`plot::Circle2d(r, [t,r], t = 0..3*PI),`  
`Scaling=Constrained)`



Ciclóide alongada:  $a = 1.4$   $r = 1$

Fig. 3.3-7

## Epiciclóide

- `reset() //Epiciclóide`  
`a:=1.0:r:=1.0:R:=4.0://a=r e R:r = 4`  
`plot(`  
`plot::Curve2d([ (R+r)*cos(t)-a*cos(t*(R+r)/r),`  
`(R+r)*sin(t)-a*sin(t*(R+r)/r)],t=0..tmax, tmax = 0..2*PI),`  
`plot::Point2d([ (R+r)*cos(t) - a*cos(t*(R+r)/r),`  
`(R+r)*sin(t)-a*sin(t*(R+r)/r)], t = 0..2*PI,`  
`PointSize = 2*unit::mm,Color=RGB::Red),`  
`plot::Line2d([ (R+r)*cos(t),`  
`(R+r)*sin(t)], [(R+r)*cos(t)-a*cos(t*(R+r)/r),`  
`(R+r)*sin(t)-a*sin(t*(R+r)/r)],`  
`Color=RGB::Green,t = 0..2*PI),`  
`plot::Circle2d(R, [0,0],Color=RGB::Black, t = 0..2*PI),`  
`plot::Circle2d(r, [(R+r)*cos(t),`  
`(R+r)*sin(t)],Color=RGB::Black, t = 0..2*PI),`  
`Scaling=Constrained)`

Epicycloide normal:  $a = r = 1$ ,  $R:r = 4$




Fig. 3.3-8

Epicycloide alongada:  $a = 1.5$ ,  $r = 1$ ,  $R:r = 3$




Fig. 3.3-9

## Hipociclóide

- ```

reset()://Hipociclóide
a:=1.5:r:=1.0:R:=4.0://a>r e R:r = 4
plot(
plot::Curve2d([(R-r)*cos(t)+a*cos(t*(R-r)/r),
(R-r)*sin(t)-a*sin(t*(R-r)/r)],t=0..tmax, tmax = 0..2*PI),
plot::Point2d([(R-r)*cos(t) + a*cos(t*(R-r)/r),
(R-r)*sin(t)-a*sin(t*(R-r)/r)], t = 0..2*PI,
PointSize = 2*unit::mm,Color=RGB::Red),
plot::Line2d([(R-r)*cos(t),
(R-r)*sin(t)],[(R-r)*cos(t)+a*cos(t*(R-r)/r),
(R-r)*sin(t)-a*sin(t*(R-r)/r)],
Color=RGB::Green,t = 0..2*PI),
plot::Circle2d(R, [0,0],Color=RGB::Black, t = 0..2*PI),
plot::Circle2d(r, [(R-r)*cos(t),
(R-r)*sin(t)],Color=RGB::Black, t = 0..2*PI),
Scaling=Constrained)

```

Hipociclóide alongada: $a=1.5$, $r = 1$, $R:r = 4$




Fig. 3.3-10

Asteróide (= hipociclóide normal com $4r = R$)




Fig. 3.3-11

3.3.3 Solução com lápis e papel

Agora vamos falar um pouco sobre o fundo teórico da família das curvas até aqui traçadas. As equações (3.3-1) são uma representação paramétrica de uma cicloide para o caso $a = r$. Nas animações temos usado a seguinte forma geral:

$$x(\theta) = r\omega\theta - a \cdot \text{sen}\omega\theta$$

$$y(\theta) = r - a \cdot \text{cos}\omega\theta$$

ω , a velocidade angular, foi, em nossos exemplos, sempre de 1 radiano por segundo.

A fig. 3.3-1 mostra que $PC = BC$, e daí temos que o ângulo $PBC = \text{ângulo } BPC$. Analisando os ângulos no triângulo PBC , vemos que $\tau = (\pi - \theta)/2$.

O ponto P participa simultaneamente de duas velocidades: uma tangencial ao círculo rolando e uma horizontal ao eixo- x .

O módulo de ambas as velocidades é $v = r \cdot \omega = r \cdot d\phi/dt$.

O vetor resultante das velocidades passa pelos pontos P e B , ou seja, ele coincide com a tangente à cicloide em P .

Se colocarmos na equação da tangente, $y = mx + n$, as coordenadas dos pontos P ($r(\theta - \text{sen}\theta)$, $r(1 - \text{cos}\theta)$) e B ($r\theta$, $2r$), obteremos para o coeficiente angular da reta tangente (a declividade) m a expressão $m = (1 + \text{cos}\theta)/\text{sen}\theta = \text{ctg}(\theta/2)$. Ou seja,

$$m := \text{tg } \tau = \text{ctg}(\theta/2) \quad (3.3-4)$$

Vemos, assim, que os ângulos θ (théta) e τ (tau) estão intimamente relacionados.

É bastante instrutivo a dedução dessa equação por meio da definição $m = \text{tg}\tau = dy/dx$:

Vejamos : No caso $a = r$, obtemos das equações 3.3-1

$$dx/d\theta = x'(\theta) = r - r \cdot \text{cos}\theta = r(1 - \text{cos}\theta) \quad \text{e} \quad dy/d\theta = y'(\theta) = r \cdot \text{sen}\theta$$

Daí obtemos $dy/dx = dy/d\theta \cdot d\theta/dx = \text{sen}\theta/(1 - \text{cos}\theta) = \text{ctg}(\theta/2)$, entyo, outra vez (3.3-4).

Para calcular o **comprimento** de um arco da cicloide (θ de 0 até 2π), precisamos do elemento de arco, $ds = (dx^2 + dy^2)^{1/2}$. Substituindo aqui dx e dy por $dx = r(1 - \text{cos}\theta)d\theta$ e $dy = r\text{sen}\theta d\theta$, obtemos para o elemento de arco a formula

$$ds = 2r \cdot \sin(\theta/2) \cdot d\theta \quad (3.3-5)$$

Integrando (3.3-5), obtemos para o comprimento de um arco inteiro da cicloide o valor

$$s = 2r \int_0^{2\pi} \sin(t/2) dt = -4r[\cos(t/2)]_0^{2\pi} = 8r$$

Um arco inteiro da cicloide simples ($a = r$) tem, então, um comprimento que é quatro vezes o diâmetro do círculo geratriz. Observe, que π não aparece no resultado!

A área entre esse arco da cicloide e do eixo-x é três vezes a área do círculo rolante:

$$A = \int_{\phi=0}^{2\pi} y dx = r^2 \int_0^{2\pi} (1 - \cos t - \cos^2 t) dt = 3\pi r^2$$

onde usamos a transformação: $y \cdot dx = y(t) \cdot x'(t) \cdot dt = r^2 \cdot (1 - \cos t)^2$

Devera ser fácil, para o MuPAD, de calcular essas integrais. Efetivamente, obtemos de modo ultra-rápido:

Comprimento:

- `f:= 2*r*sin(t/2):`
`comprimento:=int(f,t=0..2*PI)`

`8.r`

Área:

- `f:=r^2*(1-cos(t))^2:`
`area:= int(f,t=0..2*PI)`

`3*pi*r^2`

ou em forma mais detalhada:

$$y := r * (1 - \cos(t)) :$$

$$x := r * (t - \sin(t)) :$$

$$\text{area} := \int (y * \text{diff}(x, t), t=0..2*PI)$$

No caso da **ciclóide invertida** temos, em vez de $\tau = \text{ctg}(\theta/2)$, a relação:

$\tau = \text{tg}(\theta/2)$, que implica $\tau = \theta/2$ e $ds = 2r \cdot \cos(\theta/2) d\theta$.

O comprimento do arco definido por 0 e θ está dada por a seguinte integral:

$$s = \int_0^{\theta} 2r \cos\left(\frac{t}{2}\right) dt = 4r \cdot \text{sen}\left(\frac{\theta}{2}\right) \quad (3.3-6)$$

Na equação (3.3-3) apareceu o raio de curvatura ρ (letra grega "RO"). Para explicar este termo, temos de calcular a derivada $ds/d\tau$:

$$\rho := ds/d\tau = ds/d\theta \cdot d\theta/d\tau = 2r \cdot \cos(\theta/2) \cdot 2 = 4r \cdot \cos\theta$$

A grandeza $\rho := |ds/d\tau|$ é, por definição, o raio de curvatura da trajetória num ponto dado.

O círculo de raio ρ (**círculo osculador**, veja o capítulo 3.4) é o círculo que melhor se ajusta a uma curva num ponto dado. Se a curva for um círculo, ρ será igual ao raio dele.

O valor recíproco de ρ é a curvatura $\kappa = 1/\rho$ (kápa) de uma curva plana. Ela mede a alteração da direção por unidade de distância ao longo da curva. No **movimento circular**, temos para a aceleração normal (ou centrípeta) a expressão $\mathbf{a}_n = v^2/r$, mas, no caso de uma trajetória não circular, temos de substituir r por ρ , veja também a equação 3.3-3.

Calculemos, agora, o **período de um pêndulo cicloidal**. Para isso precisamos primeiramente uma expressão para a velocidade (e com isso a energia cinética).

Partimos de

$$\frac{dv}{dt} = \frac{dv}{ds} \cdot \frac{ds}{dt} = \frac{dv}{ds} v = \frac{d(v^2/2)}{ds} \quad (3.3-7)$$

Da equação (3.3-2), $mdv/dt = -mg \cdot \text{sen} \tau$, obtemos $dv/dt = -g \cdot \text{sen} \tau = -g/(4r) \cdot s$, utilizando a equação (3.3-6).

A integração de $d(v^2)/ds = -g/(2r) \cdot s$ dá

$$v^2 = -g/(4r) \cdot s^2 + C = -4rg \cdot \text{sen}^2 \tau + C \quad (3.3-8)$$

A partícula começa a mover-se no ponto onde $\tau = \tau_0$. A velocidade, neste ponto, seja zero. Se a partícula parte no ponto mais alto A, veja fig. (3.3-5), temos $\theta = \pi$ e $\tau = \theta/2 = \pi/2$. No ponto mais baixo temos $\tau = 0$. Com isso está determinada a constante de integração C, e obtemos, finalmente,

$$v^2 = 4rg(\text{sen}^2 \tau_0 - \text{sen}^2 \tau) \quad (3.3-9)$$

v depende, então, da posição da massa. Façamos $\tau_0 = \pi/2$ e calculemos $T = 4 \int ds/v$:

$$T = 4 \int_0^{\pi/2} \frac{4r \cos \tau d\tau}{\sqrt{4rg \cos \tau}} = 2\pi \sqrt{\frac{4r}{g}} \quad (3.3-10)$$

Vemos que o período do pêndulo cicloidal é independente da amplitude.

Essa expressão é válida para qualquer amplitude, pois não foi feita nenhuma aproximação com relação ao ângulo θ . Esse resultado chama-se propriedade de isocronismo (de duração igual).

$T/4 = \pi(r/g)^{1/2}$ será o tempo que uma partícula precisa para chegar ao ponto mais baixo da cicloide. Entre todas as curvas que conectam o ponto mais alto com o ponto mais baixo não existe nenhuma mais rápida. O tempo para chegar ao ponto mais baixo será maior do que o dado por $\pi(r/g)^{1/2}$ (= 1s para $r = 1$ m e $g = 9.81$ m/s²). Um corpo que desce sem atrito um plano inclinado está sujeito as forças **N** e **mg**. A componente do peso $mg \cdot \text{sen}(\alpha)$, paralela ao plano, produz a aceleração $a = g \cdot \text{sen}(\alpha)$. Utilizando isso e $v = a \cdot t$ e $v = (2gh)^{1/2}$, obtemos para o tempo de descida

$$t = \frac{\sqrt{2gh}}{g \cdot \text{sen}(\alpha)}$$

Com os valores de $h=2r=2$ m, $\text{tg}(\alpha) = 2/\pi$ e $g = 9.81$ m/s² obtemos $t = 1,189$ s, ou seja, 0,186 segundos mais do que a descida ao longo da cicloide.

